

Proyecto Ventas

Hecho por: Juan Gómez Ruiz

Introducción

En este documento recopilaremos los distintos métodos de arrays variables utilizados en el proyecto de ventas donde deberemos de registrar productos, ventas, etc.

Menú del proyecto

Para comenzar deberemos de mostrar el menú por el cual navegaremos por cada una de sus opciones:

```
Escribir "=== Proyecto Ventas ==="

Repetir
    Escribir "1-- Registrar productos y precios"
    Escribir "2-- Registrar ventas diarias"
    Escribir "3-- Mostrar tabla completa de ventas"
    Escribir "4-- Calcular total vendido por producto"
    Escribir "5-- Calcular total vendido por día|"
    Escribir "6-- Mostrar producto más y menos vendido"
    Escribir "7-- Buscar ventas de un producto específico"
    Escribir "8-- Salir del sistema"
    Escribir "Elija una opción:"
    Leer opciones
```

En este podemos observar un total de 8 opciones, la última opción la más importante ya que es con la cual deberemos de salir del programa, es decir si introducimos un 8 el programa acabará. También debajo del menú encontramos una pequeña cuestión que se le pregunta al usuario para pedir que introduzca un número que se encuentre dentro del menú.

Opción 1

```
Segun opciones Hacer
1:
  Escribir "¿Cuántos productos quiere introducir?"
  Leer producto
  Dimension productos[producto], precio[producto]
  Para i ← 1 Hasta producto Con Paso 1
    Escribir "Nombre del producto ", i, ":"
    Leer productos[i]
  FinPara
  Para i ← 1 Hasta producto Con Paso 1
    Escribir "Precio de ", productos[i], ":"
    Leer precio[i]
  FinPara
```

Aquí encontraremos la opción de registro de productos y precios. Nos pedirá cuantos productos queremos introducir. Tras introducir los productos se creará el array de productos con el límite del array con el número de productos que queremos introducir. A continuación nos pedirá el nombre del producto. También crearemos por último el array de los precios de los productos previamente introducidos.

Para crear los arrays utilizaremos Dimensión que es como hemos aprendido en clase a como crear los arrays y para introducir los precios y los nombres de los productos con el para es el bucle mas optimo para este metodo ya que sabemos que el bucle va a acabar en algún momento.

Opción 2

```
2:
  Si producto = 0 Entonces
    Escribir "Primero registre productos (opción 1)."
  Sino
    Escribir "¿Cuántos días quiere registrar?"
    Leer Dias
    Dimension ventasdiarias[producto, Dias]
    Para i ← 1 Hasta Dias Con Paso 1
      Para j ← 1 Hasta producto Con Paso 1
        Escribir "Ventas de ", productos[j], " el día ", i, ":"
        Leer ventasdiarias[j, i]
      FinPara
    FinPara
  FinSi
```

En la segunda opción encontramos antes de nada encontramos una comprobación en la que si no se ha introducido ningún valor dentro de producto que mande un mensaje de error. También encontramos las distintas ventas diarias que queramos registrar. En primer lugar deberemos de ver cuantos días

queremos introducir. Crearemos el array bidimensional de ventas diarias donde almacenaremos los Días y los productos. A continuación haremos un Para, en este pararemos hasta que los días por registrar lleguen al límite y haremos otro bucle dentro para registrar el número de productos que ha vendido un producto en ese día.

Opción 3

```
3:
  Escribir "Has elegido mostrar tabla completa de ventas"
  Para i ← 1 Hasta Dias Con Paso 1
    Escribir ""
    Escribir "Día ", i
    Para j ← 1 Hasta producto Con Paso 1
      Escribir productos[j], " Precio: ", precio[j], " Vendido: ", ventasdiarias[j, i], " unidades"
    FinPara
  FinPara
```

En la tercera opción encontraremos mostrar todas las ventas de cada producto. En mi caso he decidido mostrar todas las ventas de cada día y el total vendido. En este caso volveremos a utilizar los Para concatenados para que funcione el código. En este caso al mostrar solo datos no necesitaremos crear un nuevo array

Opción 4

```
4:
  Escribir "Has elegido calcular total por producto"
  Dimension totalproductos[producto]
  Para j ← 1 Hasta producto Con Paso 1
    totalproductos[j] ← 0
    Para i ← 1 Hasta Dias Con Paso 1
      totalproductos[j] ← totalproductos[j] + ventasdiarias[j, i] * precio[j]
    FinPara
  FinPara
  Para j ← 1 Hasta producto Con Paso 1
    Escribir productos[j], ": ", totalproductos[j]
  FinPara
```

En la 4ª opción deberemos de calcular el total de los productos vendidos. Para ello he creado un array que contiene todos los productos que hemos introducido. A continuación he creado dos Para concatenados donde he definido el total de los productos vendidos a 0 y a continuación en la siguiente estructura he definido que el total de productos vendidos sea la suma de las ventas diarias (opción 2) por el precio de cada producto.

Una vez calculado el total de todos los productos vendidos he creado otra estructura para poner el resultado total de los productos vendidos. En este caso será el array que hemos creado previamente.

Opción 5

```
5:
  Escribir "Has elegido calcular total por día"
  Para i ← 1 Hasta Dias Con Paso 1
    totalDia ← 0
    Escribir "Día ", i, ":"
    Para j ← 1 Hasta producto Con Paso 1
      ganancia ← ventasdiarias[j, i] * precio[j]
      totalDia ← totalDia + ganancia
      Escribir "  ", productos[j], ": ", ganancia
    FinPara
    Escribir "Total del día ", i, ": ", totalDia
  FinPara
```

En la 5ª opción está la opción de calcular el total por día . Para esta opción no nos hace falta definir un array. En mi caso he vuelto a poner dos estructuras Para concatenados entre sí donde he definido la variable total Día a 0. Dentro de la otra estructura he creado otra variable llamada ganancia donde he almacenado el resultado de las ventas diarias por el precio de los productos, después a la variable de total Día le he sumado la ganancia. Total Día la he utilizado como variable que almacena el total de cada Día. Al final esta la línea donde nos indicará la ganancia de cada día.

Opción 6

```
6:
Escribir "Has elegido Mostrar Producto más y menos vendido"
Dimension totalUnidades[producto]
Para j ← 1 Hasta producto Con Paso 1
    totalUnidades[j] ← 0
    Para i ← 1 Hasta Dias Con Paso 1
        totalUnidades[j] ← totalUnidades[j] + ventasdiarias[j, i]
    FinPara
FinPara

maxVenta ← totalUnidades[1]
minVenta ← totalUnidades[1]
masVendido ← productos[1]
menosVendido ← productos[1]

Para j ← 2 Hasta producto Con Paso 1
    Si totalUnidades[j] > maxVenta Entonces
        maxVenta ← totalUnidades[j]
        masVendido ← productos[j]
    FinSi
    Si totalUnidades[j] < minVenta Entonces
        minVenta ← totalUnidades[j]
        menosVendido ← productos[j]
    FinSi
FinPara

Escribir "Producto más vendido: ", masVendido, " con ", maxVenta, " unidades."
Escribir "Producto menos vendido: ", menosVendido, " con ", minVenta, " unidades."
```

En la 6ª opción encontramos de primeras la definición del array del total de las unidades. A continuación creó una estructura Para en la que al array que he creado previamente le sumo las ventas diarias de los productos.

A continuación defino 4 variables, 2 para los productos más vendidos y otro para los menos vendidos. A continuación creo otra estructura Para en la que esta vez la inicializo la j a 2 (Debido a que las variables tienen el valor 1) y a continuación compruebo si el valor siguiente es mayor a los valores iniciales. Pasa lo mismo con los valores mínimos.

Y por último deberá de aparecer tanto el producto menos vendido y el más vendido.

Opción 7

```
7:
...
Escribir "Ingrese el nombre del producto a buscar:"
Leer buscarProducto
encontrado ← Falso
Para j ← 1 Hasta producto Con Paso 1
...
Si productos[j] = buscarProducto Entonces
...
    encontrado ← Verdadero
    Escribir "Ventas de ", productos[j], ":"
    Para i ← 1 Hasta Dias Con Paso 1
    ...
        Escribir "    Día ", i, ": ", ventasdiarias[j, i], " unidades."
    FinPara
...
FinSi
FinPara
Si No encontrado Entonces
    Escribir "No se encontró el producto ", buscarProducto
FinSi
```

Y por último en la 7ª opción encontraremos la búsqueda del producto por su nombre. En mi caso he decidido crear un boolean. Cuando se lee el producto se buscará mediante un "Para " en toda la lista de los productos. Si se encuentra el producto entonces se hará un pequeño menú donde se mostrarán mediante otra estructura Para los días y las ventas diarias de dicho producto. En el caso de que no se encuentre dicho producto entonces mostrará un mensaje de error.

Opción 8

```
8:
...
Escribir "Saliendo del programa..."
```

Y por último la 8ª opción. En el caso de que se introduzca esta opción mostrará este mensaje de salida y el programa terminará.

Cosas que podría haber usado

Una de las cosas que podría haber utilizado es el mientras para el menú en vez de repetir mientras. también podríamos utilizar el mientras para guardar por ejemplo los valores de los productos pero en mi opinión la estructura para es la mas optima para este tipo de ejercicios.

Principalmente utilizo el para ya que es la manera más óptima para este tipo de proyecto ya que siempre sabemos la cantidad de valores que vamos a introducir cada vez y los otros tipos de bucles nos pueden servir pero no van a ser los más óptimos.